Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 11
Functions of Several Variables

Initializations

```
> restart;
  with(plots):
  setoptions3d(axes=boxed);
>
```

11.1 Definition and Plotting of Functions of two Variables

Examples

Example 11.1.1

Define $f(x, y) = x^2 \sin y$ as a Maple function and sketch its graph.

Solution

```
> f:=(x, y)->x^2*sin(y);
```

$$f := (x, y) \to x^2 \sin(y) \qquad \textbf{(2.1.1.1)}$$

```
> plot3d(f(x, y), x=-2..2, y=-10..10, style=patch, labels=
  [x,y,z], grid=[20,80]);
```



>

Example 11.1.2

Sometimes graphing routines have difficulties with surfaces that have vertical tangent planes. For instance try to plot the graph of $f(x,y) = \sqrt{5-2x+3y}$.

Solution

Code the function and create the graph.

```
> f:=(x,y)->sqrt(5-2*x+3*y);
```

$$f := (x,y) \to \sqrt{5-2x+3y} \qquad \textbf{(2.1.2.1)}$$

```
> plot3d(f(x,y), x=-3..3, y=-3..3, orientation=[-20,60],
  labels=[x,y,z]);
```



A way to remedy this problem is to parametrize this surface in such a way that one of the parameters is constant along the line $5-2x+3y=0$, where the vertical tangent plane occurs. For instance let $s = 5-2x+3y$ and use x and s as parameters.

```
> eq:=op(1, f(x,y))=s;
  rr:=isolate(eq, y);
```

$$eq := 5-2x+3y = s$$

$$rr := y = \frac{1}{3}s - \frac{5}{3} + \frac{2}{3}x \qquad \textbf{(2.1.2.2)}$$

```
> plot3d(subs(rr, [x,y,f(x,y)]), x=-3..3, s=0..10,
  orientation=[-20,60], labels=[x,y,z]);
```



Example 11.1.3

Define $g(x, y, z) = x + y^2 + z^3$ as a Maple function and evaluate $g(\sqrt{2}, \pi, e)$.

Solution

```
> g:=(x, y, z)->x+y^2+z^3;
```

$$g := (x, y, z) \to x + y^2 + z^3 \qquad \textbf{(2.1.3.1)}$$

```
> g(sqrt(2), Pi, exp(1));
```

$$\sqrt{2} + \pi^2 + (e)^3 \qquad \textbf{(2.1.3.2)}$$

```
>
```

11.2 Contour Plots and Level Curves

In stead of a surface plot, like the one produced in Example 10.1.1 for the function $f(x, y) = x^2 \sin(y)$, we can produce a set of level curves. A level curve is the projection on the XOY-plane of a path on the surface $z = f(x, y)$ with constant z-coordinate. The Maple syntax for producing level curves is **contourplot** or **contourplot3d**. Both routines are located in the **plots** package.

Examples

Example 11.2.1

Create a contourplot for the function $f(x, y) = x^2 \sin(y)$, first by using contourplot, then by using contourplot3d.

Solution

```
> f:=(x, y)->x^2*sin(y);
```

$$f := (x, y) \to x^2 \sin(y) \qquad \textbf{(3.1.1.1)}$$

```
> contourplot(f(x, y), x=-2..2, y=-10..10, numpoints=3600);
```



The advantage of using **contourplot3d** instead of **contourplo**t is that the **contourplot3d** allows for more detailed color control. The color control allows the user to identify high levels (red) and low levels (purple).

```
> contourplot3d(f(x, y), x=-2..2, y=-10..10, shading=ZHUE,
  axes=boxed, orientation=[-90, 0], numpoints=2500);
```



A similar picture can be obtained by using the **plot3d** command with a **style=patchcontour** option.

```
> plot3d(f(x, y), x=-2..2, y=-10..10, shading=ZHUE, style=
  patchcontour, axes=boxed, orientation=[-90, 0],
  numpoints=2500);
```


10
5
0
-5
-10
y
-2
-1
0
1
2
x